

A COMPLETE FORMALIZATION OF FERMAT'S LAST THEOREM FOR REGULAR PRIMES IN LEAN

ALEX J. BEST, CHRISTOPHER BIRKBECK, RICCARDO BRASCA, ERIC RODRIGUEZ BOIDI, RUBEN VAN DE VELDE, AND ANDREW YANG


ABSTRACT. We formalize a complete proof of the regular case of Fermat's Last Theorem in the LEAN4 theorem prover. Our formalization includes a proof of Kummer's lemma, that is the main obstruction to Fermat's Last Theorem for regular primes. Rather than following the modern proof of Kummer's lemma via class field theory, we prove it by using Hilbert's Theorems 90-94 in a way that is more amenable to formalization.

1. INTRODUCTION

For $x, y, z \in \mathbb{Z}$ and $n \in \mathbb{N}$ with $n > 2$, Fermat's Last Theorem (FLT) is the result that there are no solutions to

$$x^n + y^n = z^n$$

with x, y and z all different from 0. This apparently easy result has been stated by Pierre de Fermat around 1637, but the first full proof, by Andrew Wiles and Richard Taylor, only appeared in 1995 in the two papers [Wi195, TW95]. The quest for a proof of FLT has a long history of driving the development of new mathematics and this has continued now as mathematicians and computer scientists look to formalize mathematics. The importance of FLT lies not in the theorem itself (as number theory is full of seemingly easy equations that turn out to be extremely difficult to solve), but in the theories that has been developed to solve it. Indeed, one can say that the whole field of algebraic number theory originated in attempting to prove FLT.

An elementary argument shows that it is enough to prove FLT in the case the exponent is a prime number $p \neq 2$. In what follows we will describe the formalization (in the LEAN4 theorem prover) of FLT in the special case where the exponent p is what is known as a regular prime. That is, p is a prime number (different from 2) that does not divide the class number of the p -th cyclotomic extension $\mathbb{Q}(e^{\frac{2\pi i}{p}})/\mathbb{Q}$. This case is significantly simpler than the full version and it is amenable to formalization given the current state¹ of our formalized libraries, such as MATHLIB [mC20]. There are several reasons for wanting to formalize the regular case, firstly as a test of MATHLIB. Even though the tools we need here are much more basic than those required for the full proof, by formalizing this case we have extensively developed the number theory library, finding many issues with our original implementations which as a result of this work have now been addressed. Secondly, many of the tools required are of independent interest, such as discriminants, cyclotomic fields, ramification results, etc, all of which were originally formalized as a part of this work. Having FLT-REGULAR as our goal allowed us to develop these theories in a cohesive manner. Lastly, this represents the first formalization of a non-trivial family of cases of FLT (a family that is conjecturally infinite). Throughout the paper, we use the symbol  for external links. Almost every mathematical statement and definition will be accompanied by such a link directly to the source code for the corresponding statement in MATHLIB or in FLT-REGULAR. To keep the links usable, they are all to a fixed commit of the master branch (the most recent one at the time of writing).

In the previous work [BBBRB23] of (some of) the authors, we formalized the following first step towards proving FLT in the regular case:

Theorem 1.1 (Case 1). *Let p be an odd regular prime. Then $x^p + y^p = z^p$ has no solutions with $x, y, z \in \mathbb{Z}$ and $\gcd(xyz, p) = 1$.*

Here we will be concerned with the second case, which replaces the condition that $\gcd(xyz, p) = 1$ with $xyz \neq 0$. The main difficulty in this is the need for Kummer's Lemma, which states that:

¹A partial formalization of the full proof is a [work in progress](#) by Kevin Buzzard, and it will likely take several years to complete.

Theorem 1.2. *Let p be an odd regular prime and let $\zeta \in \mathbb{C}$ be a primitive p -th root of unity. If $u \in \mathbb{Z}[\zeta]^\times$ is a unit such that $u \equiv a \pmod{p}$ for an integer a , then there exists $v \in \mathbb{Z}[\zeta]^\times$ such that $u = v^p$.*

While there are several proofs of this result, many make use of class field theory or the p -adic class number formula, which are beyond what is currently available in MATHLIB. For this reason, we have instead opted for a more ‘elementary’ proof, requiring only some basic results about group cohomology, specifically Hilbert’s Theorems 90-94.

Here is an outline of the paper. In Section 2 we fix the informal and formal notation that will be used through the paper, explaining the very first issues that appear in the formalization process. In Section 3, we recall how in [BBBRB23] we formalized Theorem 3.1, reducing FLT to Theorem 4.1. In Section 4 we reduce the proof of Case 2 to Kummer’s lemma, Theorem 5.1. This is mathematically not very difficult, but the proof is rather intricate, and its formalization is quite challenging. We then move on to the heart of our work, the formalization of Kummer’s lemma. In Section 5, we prove Kummer’s lemma *assuming Hilbert’s Theorem 94*, a cohomological result that is historically a precursor of global class field theory. In particular, the only thing that remain to be proven is Theorem 6.1: if L/K is an unramified extension of number fields of odd prime degree, then $[L : K]$ divides the class number of K . Starting with Section 6, our work is basically independent of FLT, and only concerns the cohomology of number fields. We then reduce Theorem 6.1 to Theorem 6.2, and the latter to Hilbert’s Theorems 92 and 90, Theorems 7.1 and 8.1. In Section 7 we prove Hilbert’s Theorem 92, which is the hardest part of our formalization. The proof we formalize is a simplification of Hilbert’s original proof, but it is still subtle and requires a lot of care. In particular it depends on nontrivial results about the units of the ring of integers of number fields. In Section 8 we explain the formalization of the version of Hilbert’s Theorem 90 that we need, using what is already in MATHLIB. This finishes the proof of Kummer’s lemma, and hence the formalization of FLT for regular primes. In Section 9 we discuss some of the issues we encountered in the formalization process, and how we solved them. Finally, in Section 10 we discuss some future work.


As it is clear from the above discussion, the paper follows a backwards reasoning: we start from the final result we want to prove, and we work our way back to the results we need. This reflects our formalization process: the final goal, FLT for regular primes, was clear from the beginning and the statement was very easy to formalize. Also the fact that we needed to split the proof in the two cases was clear, as all the known approaches follow this strategy. Once Case 1 was done, we moved to Case 2, and again all known approaches use Kummer’s lemma. A key observation is that, even if the proof is difficult, the formalization of the statement is easy, so we decided to finish the work assuming Kummer’s lemma. Now, as there are several possible ways of proving Theorem 5.1 that involve various complicated objects (cohomology groups, L -functions etc.) it was key to chose an approach that could be formalised in a reasonable time-frame. Working backwards, we were sure that we were really making progress at any step of the formalization (of course there is always the risk of ending up with something very hard to prove, but in this way we minimize the risk of formalizing useless *definitions*). This is a very different strategy with respect to projects like the [Liquid Tensor Experiment](#) where the *statement* of the result is already difficult to formalize, not only its proof.

2. INFORMAL AND FORMAL NOTATION

Throughout the paper, p will be an odd regular prime. Although FLT is a statement about the integers, the proof we are formalizing (as all modern approaches) requires a significant input from algebraic number theory. Here is our basic setup. Let $\mu_p \subseteq \mathbb{C}$ be the set of p -th roots of unity. We fix $\zeta \in \mu_p$ a primitive root of unity and we will write K for the number field $\mathbb{Q}(\zeta)$. By definition of regularity, p does divide the class number of K . The ring of integers of K will be denoted \mathcal{O}_K : we have that $\mathcal{O}_K = \mathbb{Z}[\zeta]$ (a fact that we formalized in [BBBRB23]).

As is customary in formalization projects, it is better to avoid working directly with the field $\mathbb{Q}(\zeta)$ (that would be `Algebra.adjoin ℚ {ζ}`), otherwise all our results would apply only to that extension of \mathbb{Q} , and not to any extension that is abstractly isomorphic to it. Instead, we work with any p -th cyclotomic extension of \mathbb{Q} , as follows.

```
variable {p : ℕ+} {K : Type*} [Field K] [NumberField K]
[IsCyclotomicExtension {p} ℚ K]
```

The class `IsCyclotomicExtension` , that is a fundamental prerequisite for our formalization, has been introduced into MATHLIB during the first steps of our project (see [BBBRB23]). The instances `[Field K]` and `[NumberField K]` ensure that K is an extension of \mathbb{Q} that is generated by a primitive p -root of

unity. We sometimes use `[CharZero K]` instead of `[NumberField K]` since some results in MATHLIB use this formulation (the two hypothesis are equivalent in our context). Note that here p is of type $(p : \mathbb{N}^+)$, which is the type of positive natural numbers and in particular it is *not* a natural number. This causes mild annoyances as we are forced to use the coercion $(\uparrow p : \mathbb{N})$ to see p as a natural number or the fact that a prime number is positive to move back and forth. This is because `IsCyclotomicExtension` takes as first argument a set $(s : \text{Set } \mathbb{N}^+)$ (this is a design decision made at the beginning of the formalization of the theory of cyclotomic extensions in MATHLIB, and our work here has highlighted the need for it to be refactored in the future).

Instead of fixing a primitive p -th root of unity ζ once and for all, it is more convenient in LEAN to work with an unspecified root primitive root.

```
variable {ζ : K} (hζ : IsPrimitiveRoot ζ p)
```

One issue that appears, is that now $(\zeta : K)$ has type K , and not \mathcal{O}_K or \mathcal{O}_K^\times , forcing us to use several coercions. To make the formalization process as smooth as possible, we usually consider $(h\zeta.\text{unit}' : \mathcal{O}^\times)$, that is the same as ζ , but seen as a unit of the ring of integers. After having set the relevant `@[simp]` and `@[norm_cast]` tags, we can use various tactics like `push_cast` or `field_simp` to avoid mathematically irrelevant annoyances.

The ring of integers of K is denoted by \mathcal{O}_K [↗](#), and MATHLIB contains an extensive library of results about it. For example, the fact that \mathcal{O}_K is a Dedekind domain [↗](#) and the fact that $\mathcal{O}_K = \mathbb{Z}[\zeta]$ [↗](#) are both formalized: the former in the work [\[BDNdC22\]](#). To state that p is a prime, we use `[Fact p.Prime]` (here $(p : \mathbb{N})$): indeed `Nat.Prime` is not a class, but using `Fact` we can record primality as an instance and use typeclass inference. Since regularity is a property of a natural number, in order to define regularity we cannot use the field K fixed above. We use instead `CyclotomicField p ℚ`: a fixed model of a p -th cyclotomic extension of \mathbb{Q} . Here is our definition of being a regular prime [↗](#).

```
/-- A natural number 'n' is regular if 'n' is coprime with the cardinal of the class group -/
def IsRegularNumber (n : ℕ) [hn : Fact (0 < n)] : Prop :=
  n.Coprime <| Fintype.card <| ClassGroup (0 <| CyclotomicField (n, hn.1) ℚ)

def IsRegularPrime (p : ℕ) [Fact p.Prime] : Prop := IsRegularNumber p
```

Note that we define the notion of being regular for any positive integer as being coprime with the class number of the corresponding cyclotomic extension, but this is used in practice only for prime numbers: in particular we do not claim that Kummer's lemma holds for regular numbers, it is for example false for $n = 4$. (This is customary in formalized mathematics, since primality is not needed to state the condition, we do not assume it, even if this generality is not mathematically interesting.) Our main theorem is then [↗](#).

```
/-- Fermat's last theorem for regular primes. -/
theorem flt_regular {p : ℕ} [Fact p.Prime] (hreg : IsRegularPrime p) (hodd : p ≠ 2) :
  FermatLastTheoremFor p := by ...
```

Here `FermatLastTheoremFor p` [↗](#) is the statement, existing in MATHLIB,

```
∀ a b c : ℕ, a ≠ 0 → b ≠ 0 → c ≠ 0 → a ^ p + b ^ p ≠ c ^ p
```

3. CASE 1

The proof of FLT in the regular case is split into two cases. The so called “first case” of FLT is the following

Theorem 3.1 (Case 1). *Let $p \neq 2$ be a regular prime. Then $x^p + y^p = z^p$ has no solutions with $x, y, z \in \mathbb{Z}$ and $\gcd(xyz, p) = 1$.*

Here is the formalized statement [↗](#)

```
theorem caseI {a b c : ℤ} {p : ℕ} [Fact p.Prime] (hreg : IsRegularPrime p)
  (caseI : ¬(∃ p | a * b * c)) : a ^ p + b ^ p ≠ c ^ p := ...
```

The formalization of this result was completed on October 13th 2022, a little less than one year later than the starting date of the project. This work is described in [BBBBRB23], and we will only briefly list the main bits we formalized:

- (1) Definitions of cyclotomic fields and cyclotomic extensions.²
- (2) Definitions of norms, trace and discriminant for number fields.
- (3) Computing the discriminant and ring of integers for p -power cyclotomic fields.
- (4) Specific results about units in these cyclotomic field. For example that for $p \neq 2$ a prime, every unit $u \in \mathbb{Z}[\zeta_p]^\times$ can be written as $u = x\zeta_p^n$ for some $n \in \mathbb{Z}$ and $x \in \mathbb{Z}[\zeta_p]^\times$ such that $x \in \mathbb{R}$.
- (5) The proof of FLT for $n = 3$ which was done independently by one of us (RVdV).

Our formalization was entirely in LEAN3, and shortly after the end of this task, the transition from LEAN3 to LEAN4 started. Once MATHLIB moved completely to LEAN4, we started updating FLT-REGULAR. For this, we used `mathport`, a tool developed by the MATHLIB community (especially by Mario Carneiro and Gabriel Ebner) to port LEAN3 projects to LEAN4. Porting FLT-REGULAR, we encountered mainly two issues.

- Coercions (especially from \mathcal{O}_K^\times to \mathcal{O}_K to K) are omnipresent in our project, see [BBBBRB23, Section 3]. LEAN4 handles coercions in a completely different way from LEAN3, so part of the code had to be rewritten by hand. Although tedious, the same issue had appeared in porting MATHLIB, so the solutions were well understood.
- The way typeclass inference handles multiple inheritance changed in LEAN4, see for example [Wie23]. This caused slowdown in MATHLIB and in FLT-REGULAR³, which we worked to solve by fixing the issues directly in MATHLIB, something that is now almost completely done (see Section 9 below for more details).


Another important aspect of our work, described in [BBBBRB23, Section 4], is integration to MATHLIB. We now have more than 200 pull requests merged that were originally in FLT-REGULAR. Notably, almost all the prerequisites needed to formalize the proof of Case 1 are now in MATHLIB (with the exception of homogenization of polynomials) and we have already started the same process for Case 2. Even if the whole proof of FLT for regular primes is unlikely to be included into MATHLIB (being very technical and specific) we expect that all prerequisites will be upstreamed. Besides improving MATHLIB itself, we saw, especially during the port, that this process reduces the work of maintaining the code, something that can be very tedious due to the continuous development/expansion of MATHLIB.

4. CASE 2

Given the formalization of Theorem 3.1 outlined in Section 3, to formalize the proof of FLT for regular primes it is enough to consider the following (which is known as Case 2 of FLT for regular primes):

Theorem 4.1 (Case 2). *Let $p \neq 2$ be a regular prime. Then $x^p + y^p = z^p$ has no solutions such that*


- $xyz \neq 0$.
- $\gcd(x, y, z) = 1$.
- $p \mid xyz$.

Our formalization is the following 

```

/-- CaseII. -/
theorem caseII {a b c : ℤ} {p : ℕ} [hpri : Fact p.Prime]
  (hreg : IsRegularPrime p) (hodd : p ≠ 2)
  (hprod : a * b * c ≠ 0) (hgcd : ({a, b, c} : Finset ℤ).gcd id = 1)
  (caseII : ↑p ∣ a * b * c) : a ^ p + b ^ p ≠ c ^ p := by ...

```

Here is the strategy of the proof that we have formalized (following [BS66, Section V.7.1]). Let x, y and z be coprime integers such that $p \mid xyz$ and $x^p + y^p = z^p$. First note that p divides exactly one of x, y, z , and by an easy change of variables we can assume that $p \mid z$. Since we have to show that there is nontrivial solution with $\gcd(x, y, z) = 1$ and $p \mid z$, it is enough to prove that there is no nontrivial solutions such that $p \nmid y$ and $p \nmid z$. The statement we are going to prove is then 

²To date these are the only explicit examples of number fields in MATHLIB.

³In the beginning we simply increased the limit of memory allocations in the project, via the `set_option maxHeartbeats` and `set_option synthInstance.maxHeartbeats` commands, and we worked with the slow code, but this was only a temporary solution.

```

lemma not_exists_Int_solution {p : ℕ} [Fact (p.Prime)] (hreg : IsRegularPrime p)
  (hodd : p ≠ 2) :
  ¬∃ (x y z : ℤ), ¬↑p | y ∧ ↑p | z ∧ z ≠ 0 ∧ x ^ p + y ^ p = z ^ p := by ...

```

In \mathcal{O}_K we have that $p = (1 - \zeta)^{p-1}\varepsilon$ for some $\varepsilon \in \mathcal{O}_K^\times$ and the norm of $\zeta - 1$ is p , in particular p divides an integer n in \mathbb{Z} if and only if $\zeta - 1$ divides n in \mathcal{O}_K . It follows that it is enough to prove

```

lemma not_exists_solution' : {p : ℕ+} [Fact (p.Prime)]
  (hreg : (p : ℕ).Coprime <| Fintype.card <| ClassGroup (℧ K))
  ¬∃ (x y z : ℧ K), ¬(hζ.unit' : ℧ K) - 1 | y ∧ (hζ.unit' : ℧ K) - 1 | z ∧ z ≠ 0 ∧
  x ^ (p : ℕ) + y ^ (p : ℕ) = z ^ (p : ℕ) := by ...

```

Note that here $(p : \mathbb{N}^+)$ (since we need cyclotomic extensions) and then we are forced to consider the coercion $(p : \mathbb{N})$ to take exponentiation. Also, `hreg` is in terms of K rather than `CyclotomicField p ℚ` (mathematically the two notions are of course equivalent): this is because we are proving results about K , so it is more convenient to work with it directly. Since ultimately we will apply our results to the case $K = \text{CyclotomicField } p \mathbb{Q}$, this causes no problems.

Write $z = p^k z_0$ with $z_0 \in \mathbb{Z}$ such that $p \nmid z_0$. It suffices to prove that the more general equation

$$x^p + y^p = \varepsilon(\zeta - 1)^{pm} z_0^p$$

has no nontrivial solutions with $x, y, z_0 \in \mathcal{O}_K$ such that y and z_0 are not divisible by $1 - \zeta$. Here $m > 0$ and $\varepsilon \in \mathcal{O}_K^\times$. The proof follows by a descent argument (which will crucially depend on Kummer's lemma) where one shows that if such a solution exists for some $m > 0$ then automatically $m > 1$ and one can construct a solution with m replaced by $m - 1 > 0$.

We now give more details about the proof, notably explaining where Kummer's lemma is needed. Assume we have $x, y, z \in \mathcal{O}_K$ and $\varepsilon \in \mathcal{O}_K^\times$ such that y and z are not divisible by $1 - \zeta$ and that we have

$$(1) \quad x^p + y^p = \varepsilon(1 - \zeta)^{p(m+1)} z^p$$

for some natural number m . Note that we wrote $m + 1$ instead of m but we are not assuming $m > 0$, so this is equivalent to the above. This formulation is more amenable to formalization as we do not need to keep around the assumption that m is positive. In what follows we will keep this choice of ζ fixed and when discussing an arbitrary p -th root of unity we will denote them by η .

Let \mathfrak{z} and \mathfrak{p} denote the ideals generated by z and $\zeta - 1$ respectively. Note that \mathfrak{p} is the only prime ideal of \mathcal{O}_K above p . Then factoring (1) as ideals we have

$$\prod_{\eta \in \mu_p} (x + \eta y) = (\zeta - 1)^{p(m+1)} \mathfrak{z}^p.$$

It follows that each $x + \eta y$ is divisible by $(\zeta - 1)$ in \mathcal{O}_K . The idea now is to divide each side by $(\zeta - 1)^p$ and then after some manipulations, show that we can create a solution with a smaller power of m on the right hand side.

For any $\eta \in \mu_p$, we let

$$q(\eta) = \frac{x + \eta y}{\zeta - 1},$$

which by the above we know lies in \mathcal{O}_K . Using this we then show that this map composed with reduction modulo \mathfrak{p} is a bijection between μ_p and $\mathcal{O}_K/\mathfrak{p}$. We denote by η_0 the unique p -th root of unity such that $q(\eta_0) \equiv 0 \pmod{\mathfrak{p}}$.

Let $\mathfrak{m} = \gcd((x), (y))$. We have that \mathfrak{m} is coprime to \mathfrak{p} and moreover \mathfrak{m} divides the ideal generated by $q(\eta)$ for any η . We write \mathfrak{c}_η for the ideal such that $\mathfrak{c}_\eta \cdot \mathfrak{m} = q(\eta)$. Then, for $\eta_1 \neq \eta_2$, we see that \mathfrak{c}_{η_1} and \mathfrak{c}_{η_2} are coprime. Moreover, each \mathfrak{c}_η is the p -th power of some ideal which we denote by \mathfrak{a}_η .

Next we see that, for all $\eta_1, \eta_2 \in \mu_p$ we have, as fractional ideals,

$$(x + \eta_1 y)/(x + \eta_2 y) = \mathfrak{c}_{\eta_1}/\mathfrak{c}_{\eta_2} = (\mathfrak{a}_{\eta_1}/\mathfrak{a}_{\eta_2})^p$$

from which it follows that $\mathfrak{c}_{\eta_1}/\mathfrak{c}_{\eta_2}$, and hence $(\mathfrak{a}_{\eta_1}/\mathfrak{a}_{\eta_2})^p$, is principal. But now, since p is regular we have that $\mathfrak{a}_{\eta_1}/\mathfrak{a}_{\eta_2}$ is principal. Furthermore, since $\mathfrak{a}_{\eta_0} = \mathfrak{p}^m \mathfrak{a}_0$ for some ideal \mathfrak{a}_0 , we have that $\mathfrak{a}_\eta/\mathfrak{a}_0$ is principal for all $\eta \neq \eta_0$. We also have that \mathfrak{p} does not divide \mathfrak{a}_0 and this implies that actually $m \geq 1$.

Write

$$\mathfrak{a}_\eta/\mathfrak{a}_0 = \alpha_\eta/\beta_\eta$$

for $\alpha_\eta, \beta_\eta \in \mathcal{O}_K$. Then, working back through the definitions we have that there is some unit ε_η such that

$$(2) \quad \varepsilon_\eta(x + \eta_0 y)\alpha_\eta^p = (x + \eta y)(\zeta - 1)^{mp}\beta_\eta^p.$$

Now pick η_1, η_2 both different from η_0 . A simple manipulation of (2) gives

$$(\eta_2 - \eta_0)\varepsilon_{\eta_1}(\alpha_{\eta_1}\beta_{\eta_2})^p + (\eta_0 - \eta_1)\varepsilon_{\eta_2}(\alpha_{\eta_2}\beta_{\eta_1})^p = (\eta_2 - \eta_1)((\zeta - 1)^m\beta_{\eta_1}\beta_{\eta_2})^p.$$

But now note that each $\eta_i - \eta_j = (\zeta - 1)u_{ij}$ for some unit u_{ij} . Cancelling off we are then left with an equation of the form

$$\varepsilon_1 x'^p + \varepsilon_2 y'^p = \varepsilon_3 (\zeta - 1)^{pm} z'^p,$$

where $\varepsilon_1, \varepsilon_2$ and ε_3 are units of \mathcal{O}_K . Setting $u_1 = \varepsilon_1/\varepsilon_2$ and $u_2 = \varepsilon_3/\varepsilon_2$ we get

$$(3) \quad u_1 x'^p + y'^p = u_2 (\zeta - 1)^{pm} z'^p,$$

Since we already know that $m \geq 1$, we are done *if we can prove that u_1 is a p -th power*. This is the crucial step where we need Kummer's lemma, Theorem 5.1 below. Thanks to it, it is enough to show that u_1 is congruent to some integer modulo p . To prove this, note that since $(p) = \mathfrak{p}^{p-1}$, there is a unit $u \in \mathcal{O}_K^\times$ such that $pu = (\zeta - 1)^{p-1}$. Substituting in (3), we see that $p \mid u_1 x'^p + y'^p$. This implies that there is some $a \in \mathcal{O}_K$ such that u_1 is congruent to a^p modulo p . We conclude since for any $x \in \mathcal{O}_K$, we have that x^p is congruent to an integer modulo p .

With the above setup and setting $\pi = \zeta - 1$, the formalization of this argument concludes that we have (see Section 2 for an explanation of the various notations):

```
variable {K : Type*} {p : N+} [hpri : Fact p.Prime] [Field K]
[NumberField K] [IsCyclotomicExtension {p} Q K] (hp : p ≠ 2) [Fintype (ClassGroup (O K))]
(hreg : (p : N).Coprime <| Fintype.card <| ClassGroup (O K)) {ζ : K} (hζ : IsPrimitiveRoot ζ p)
{x y z : O K} {ε : (O K)×} {m : N} (hy : ¬ hζ.unit'.1 - 1 | y) (hz : ¬ hζ.unit'.1 - 1 | z)
(e : x ^ (p : N) + y ^ (p : N) = ε * ((hζ.unit'.1 - 1) ^ (m + 1) * z) ^ (p : N))

lemma exists_solution' : ∃ (x' y' z' : O K) (ε₃ : (O K)×), ¬ π | y' ∧ ¬ π | z' ∧
x' ^ (p : N) + y' ^ (p : N) = ε₃ * (π ^ m * z') ^ (p : N) --then we get a solution with m
```

5. KUMMER'S LEMMA

Kummer's lemma states the following:

Theorem 5.1. *Let p be an odd regular prime and let ζ be a primitive p -th root of unity. If $u \in \mathcal{O}_K^\times$ is a unit such that $u \equiv n \pmod{p}$ for an integer n , then there exists $v \in \mathcal{O}_K^\times$ such that $u = v^p$.*

A modern proof of Kummer's lemma goes as follows.

Proof of Theorem 5.1. Let us consider the extension $K(u^{1/p})/K$: it is a Kummer, hence abelian, extension. By a rather elementary argument (see [Was97, second proof of Theorem 5.36] or the proof at the end of this section), the extension is everywhere unramified (here is where the assumption on u is needed), so $K(u^{1/p})$ is contained in L , the Hilbert class field of $K(u^{1/p})$. By class field theory, $[L : K(u^{1/p})]$ is the class number of $K(u^{1/p})$ and in particular $p \nmid [L : K(u^{1/p})]$. It follows that the degree of $K(u^{1/p})/K$ cannot be p : the only other possibility is that the degree is 1, so $K(u^{1/p}) = K$ and u has a p -th root in K and hence in \mathcal{O}_K^\times . \square

The formalization of global class field theory is a long term project, but we are currently quite far from it, making impossible to formalize such a proof. There are other modern proofs, for example using the p -adic regulator, but again their formalization is unfeasible at the moment. For this reason we instead follow a more elementary proof relying on some basic ramification results and Hilbert's Theorems 90, 92 and 94 (see [HLA+98] for the original formulation). Those are historically the starting point of modern class field theory, so in a sense our formalized proof is the one above, but written in much more down-to-earth terms.

The formalization of the statement is the following (notations as in Section 2) \square

```
theorem eq_pow_prime_of_unit_of_congruent (u : (O K)×)
(hcong : ∃ n : Z, (p : O K) | (u - n : O K)) : ∃ v, u = v ^ (p : N) := by ...
```

Note we decided to spell out explicitly the condition that u is congruent to n modulo p , without actually using \equiv in the formal statement. This is of course equivalent, and we felt that the API was better suited to work with division (we could also have worked with ideals, but again this is more cumbersome).

We now describe with more details the proof we formalized. Let $u \in \mathcal{O}_K^\times$ and $n \in \mathbb{Z}$ such that $u \equiv n \pmod{p}$. Suppose for a moment that if $w \in \mathcal{O}_K^\times$ is a unit such that $(\zeta - 1)^p \mid w - 1$ then there exists some $v \in K$ such that $v^p = w$. Admitting this claim, we see that with our setup we have $u \equiv a \pmod{p}$ for some integer a . Now this means that $u^{p-1} \equiv 1 \pmod{p}$, which means that $u^{p-1} \equiv 1 \pmod{(\zeta - 1)^{p-1}}$ and hence $u^{p-1} \equiv 1 \pmod{(\zeta - 1)^p}$. Using the claim we then have a v' (which must necessarily be a unit) such that $v'^p = u^{p-1}$. So choosing $v = u/v'$ we see that $v^p = u$ as required.

So it remains to prove the claim, which is formalized as follows [↗](#)

```
theorem not_for_all_zeta_sub_one_pow_dvd_sub_one_of_pow_ne (u : (O K)^\times)
  (hcong : (h\zeta.unit' - 1 : O K) ^ (p : N) | (u : O K) - 1) : \!v v : K, v ^ (p : N) \neq u := by ...
```

To see this, assume for contradiction that the claim is false, so u is not a p -th power and let $L = K(u^{1/p})$. The extension $K(u^{1/p})/K$ is a Kummer extension that by our assumption has degree p . Note that there was no theory of Kummer's extensions in MATHLIB when the project began, and we developed it from the ground up. This new material is now fully integrated in the library and it is fairly complete: for example MATHLIB knows that $K(u^{1/p})/K$ is Galois with cyclic Galois group.

We now claim that $K(u^{1/p})/K$ is unramified. Again we had to develop the theory of unramified extensions from scratch, and this material is now in the process of being integrated into MATHLIB. The formal statement we use to prove unramifiedness is the following [↗](#)

```
lemma isUnramified (L) [Field L] [Algebra K L]
  [IsSplittingField K L (X ^ (p : N) - C (u : K))] : IsUnramified (O K) (O L) := by ...
```

Here L is the splitting field of $X^p - u$ over the cyclotomic field K . Consider the polynomial

$$P_u = \frac{((\zeta - 1)X - 1)^p + u}{(\zeta - 1)^p} \in K[X]$$

First of all we have that [↗](#) $P_u \in \mathcal{O}_K[X]$, it is monic [↗](#) and has degree p [↗](#). Moreover, in $K[X]$ we have that [↗](#)

$$P_u = (X - (\zeta - 1)^{-1})^p + \left(\frac{u}{\zeta - 1}\right)^p.$$

Since we are assuming that u is not a p -th power we have that [↗](#) P_u is irreducible over K . Moreover [↗](#), the roots of P_u are the

$$\alpha_m := \frac{1 - \zeta^m u^{1/p}}{\zeta - 1},$$

for $m = 0, \dots, p-1$ and so P_u is the minimal polynomial of any of the α_m over K and hence [↗](#), by Gauss's lemma, over \mathcal{O}_K . An explicit computation shows that $X - \alpha_{m_1}$ and $X - \alpha_{m_2}$ are coprime if $m_1 \neq m_2$, and thus [↗](#) P_u is a separable polynomial over \mathcal{O}_L and in particular [↗](#) it is separable over \mathcal{O}_K/I for any maximal ideal $I \subseteq \mathcal{O}_K$. Since clearly $L = K(\alpha_m)$ and α_m has separable polynomial modulo any maximal ideal of \mathcal{O}_K , we have that [↗](#) the extension L/K is everywhere unramified. Note that the proof of unramifiedness, although not very complex, is quite intricate, and it required us to develop the theory of unramified extensions. In doing so we formalized, among other things the following:

- (1) Definition of the relative different ideal $\mathfrak{d}_{K/F} := ((\mathcal{O}_K)^*)^{-1}$ where

$$M^* = \{\alpha \in K \mid \text{Tr}_{K/F}(\alpha M) \in \mathcal{O}_F\}.$$

This is now in MATHLIB [↗](#)

- (2) Proving that if K/F is an extension of number fields and S denote the set of $\alpha \in \mathcal{O}_K$ be such that $K = F(\alpha)$ then [↗](#)

$$\mathfrak{d}_{K/F} = (m'_\alpha(\alpha) : \alpha \in S)$$

where m_α is denotes the minimal polynomial of α .

- (3) If $\mathfrak{p}_F, \mathfrak{p}_K$ prime ideals in K, F respectively, with $\mathfrak{p}_K^e \parallel \mathfrak{p}_F$ for $e > 0$. Then [↗](#)

$$\mathfrak{p}_K^{e-1} \mid \mathfrak{d}_{K/F}.$$

- (4) If $K = F(\alpha)$ and m_α is separable modulo a prime ideal \mathfrak{p}_F , then [↗](#) \mathfrak{p}_F is unramified in \mathcal{O}_K .

The situation is now the following: we have a unit $u \in \mathcal{O}_K^\times$ that is not a p -th power and moreover the extension $K(u^{1/p})/K$ is everywhere unramified. We need to find a contradiction. Note that we have yet to use regularity of p . It is unlikely that one can finish the proof without some sort of class field theory result that allows one to deduce information about the class group by the existence of an abelian unramified extension. The one we formalize is Hilbert's Theorem 94, whose second part, Theorem 6.1 below, finishes the proof since by regularity p cannot divide the class number of K .

6. HILBERT'S THEOREM 94

Hilbert's Theorem 94 has two parts to it. Continuing our backwards reasoning, we start with the second one, that we used above to finish the proof of unramifiedness.

Theorem 6.1. *[Hilbert's Theorem 94, part 2] Let L/K be an unramified cyclic finite extension of number fields of odd prime degree, then $[L : K]$ divides the class number of K .*

Proof. To prove the theorem, assume that $[L : K]$ does not divide the class number and note that if I is any ideal of \mathcal{O}_K such that I_L , the extension to \mathcal{O}_L of I , is principal then I is automatically principal. Indeed, we have $\text{Norm}_{L/K}(I_L) = I^{[L:K]}$ and, I_L being principal means that $I^{[L:K]}$ is principal. Since $[L : K]$ is coprime with the class number we have that I is principal. In particular, we see that it is enough to show that there is a non-principal ideal of \mathcal{O}_K that becomes principal in \mathcal{O}_L . This is precisely the content of part 1 of Hilbert's Theorem 94, Theorem 6.2 below. \square

A formalization of the above argument is the following. We include also the proof, that is very short and follows our informal proof above. The declaration `exists_not_isPrincipal_and_isPrincipal_map` is Theorem 6.2 below.

```

theorem dvd_card_classGroup_of_isUnramified_isCyclic (K L : Type*)
  [Field K] [Field L] [NumberField K] [NumberField L] [Algebra K L]
  [FiniteDimensional K L] [IsGalois K L] [IsUnramified (O K) (O L)]
  [IsCyclic (L ≃a [K] L)] (hKL : Nat.Prime (finrank K L))
  (hKL' : finrank K L ≠ 2) :
  finrank K L | Fintype.card (ClassGroup (O K)) := by
obtain ⟨I, hI, hI'⟩ := exists_not_isPrincipal_and_isPrincipal_map K L hKL hKL'
letI := Fact.mk hKL
rw [← Int.ofNat_dvd,
  (Nat.prime_iff_prime_int.mp hKL).irreducible.dvd_iff_not_coprime,
  Nat.isCoprime_iff_coprime]
exact fun h ↦ hI (IsPrincipal_of_IsPrincipal_pow_of_Coprime _ _ h _
  (Ideal.isPrincipal_pow_finrank_of_isPrincipal_map _ hI'))

```

Note that `IsCyclic (L ≃a [K] L)` is automatic by the other assumptions, but we find it more convenient to explicitly add it. In the applications, it is found by typeclass inference, so this does not cause any issue.

We now move on to the first part of Hilbert's Theorem 94, which states that:

Theorem 6.2. *[Hilbert's Theorem 94, part 1] Let L/K be an unramified cyclic extension of number fields of odd prime degree, then there is a non-principal ideal of \mathcal{O}_K that becomes principal in \mathcal{O}_L .*

Proof. This is a consequence of Hilbert's Theorem 92, Theorem 7.1 below. Let $\eta \in \mathcal{O}_K^\times$ be a unit with $\text{Norm}_{L/K}(\eta) = 1$ and such that for all $w \in \mathcal{O}_K^\times$ we have $\eta \neq w/\sigma(w)$, where σ is a fixed generator of $\text{Gal}(L/K)$. By Hilbert's Theorem 90, Theorem 8.1 below, there is $\beta \in \mathcal{O}_K$ such that $\beta \neq 0$ and $\beta = \sigma(\beta)\eta$. We let $I := \mathcal{O}_K \cap (\beta)$, the ideal of \mathcal{O}_K given by the ideal of \mathcal{O}_L generated by β . We claim that I satisfies the required properties. Suppose for a moment that $I\mathcal{O}_L = (\beta)$. This obviously implies that the extension of I to \mathcal{O}_L is principal. If I were principal generated by γ , there would be a unit $w \in \mathcal{O}_K^\times$ such that $\beta = \gamma w$. Substituting into $\beta = \sigma(\beta)\eta$, since $\sigma(w) = w$ by $w \in \mathcal{O}_K$, we get $\eta = \gamma/\sigma(\gamma)$ that is absurd. In particular I is not principal.

We now prove that $I\mathcal{O}_L = (\beta)$. Since $\sigma^{-1}(\beta) = \sigma^{-1}(\eta)\beta$, we have that $\sigma^{-1}(I)\mathcal{O}_L = (\beta)$. Being L/K unramified, this implies that $I\mathcal{O}_L = (\beta)$ (this is a technical point where we use the unramifiedness assumption to factor $I\mathcal{O}_L$ into product of prime ideals). \square

Our formalization of the above theorem is the following.

```

theorem exists_not_isPrincipal_and_isPrincipal_map (K L : Type*)
  [Field K] [Field L] [NumberField K] [NumberField L] [Algebra K L]

```



```

[FiniteDimensional K L] [IsGalois K L] [IsUnramified (O K) (O L)] [h : IsCyclic (L ≃a [K] L)]
(hKL : Nat.Prime (finrank K L)) (hKL' : finrank K L ≠ 2) :
∃ I : Ideal (O K), ¬I.IsPrincipal ∧ (I.map (algebraMap (O K) (O L))).IsPrincipal := by
obtain ⟨σ, hσ⟩ := h
obtain ⟨η, hη, hη'⟩ := Hilbert92 hKL hKL' σ hσ
exact exists_not_isPrincipal_and_isPrincipal_map_aux σ hσ η hη (not_exists.mpr hη')

```

Here, `exists_not_isPrincipal_and_isPrincipal_map_aux` is [is](#).

```

theorem exists_not_isPrincipal_and_isPrincipal_map_aux
[IsDedekindDomain A] [IsUnramified A B] (η : B×) (hη : Algebra.norm K (algebraMap B L η) = 1)
(hη' : ¬∃ α : B×, algebraMap B L η = (algebraMap B L α)/σ (algebraMap B L α)) :
∃ I : Ideal A, ¬I.IsPrincipal ∧ (I.map (algebraMap A B)).IsPrincipal := by ...

```

The formal proof is a faithful translation of the informal one above.

7. HILBERT'S THEOREM 92

The main technical input in our formalization of Kummer's lemma is the following result:

Theorem 7.1 (Hilbert's Theorem 92). *Let K/k be a cyclic extension of number fields of odd prime degree and let σ be a generator of the Galois group. Then there exists a unit $u \in \mathcal{O}_K^\times$ such that $\text{Norm}_{K/k}(u) = 1$ and for all $\varepsilon \in \mathcal{O}_K^\times$ we have $u \neq \varepsilon/\sigma(\varepsilon)$.*

We formalize the above statement as [is](#):

```

lemma Hilbert92
[Algebra k K] [IsGalois k K] [FiniteDimensional k K] (hKL : Nat.Prime (finrank k K))
(hpodd : finrank k K ≠ 2) (σ : K ≃a [k] K) (hσ : ∀ x, x ∈ Subgroup.zpowers σ) :
∃ u : (O K)×, Algebra.norm k (u : K) = 1 ∧ ∀ ε : (O K)×, (u : K) ≠ ε / (σ ε : K) :=

```

Note that by Hilbert's Theorem 90, Theorem 8.1 below, we know that any such u must be of the form $w/\sigma(w)$ for $w \in \mathcal{O}_K$. So the content is in showing that w cannot always be chosen to be a unit. This translates to saying that a certain cohomology group does not vanish as opposed to the usual vanishing result associated with Hilbert's Theorem 90. As remarked by [SD01], the cohomological point of view does not make the proof easier, so we instead follow the classical proof described in therein, which also has the advantage of being more amenable to formalization given the current state of MATHLIB.

The constructions of a unit satisfying the requirements of Hilbert's Theorem 92 will be a consequence of the existence of a particular set of units.

7.1. Fundamental system of units.

Definition 7.2. Let σ be a generator of $\text{Gal}(K/k)$ (with K/k as in Theorem 7.1) and let r be $\text{rk}_{\mathbb{Z}}(\mathcal{O}_k^\times)$. Let $\mathcal{U}'_{K/k} = \mathcal{O}_K^\times / \mathcal{O}_k^\times$ and let $\mathcal{U}'_{K/k, \text{tors}}$ denote the torsion subgroup. Let

$$\mathcal{U}_{K/k} = \mathcal{U}'_{K/k} / \mathcal{U}'_{K/k, \text{tors}},$$

which we note is naturally a $\mathbb{Z}[\text{Gal}(K/k)]$ -module. Then a fundamental system of $r + 1$ units is a choice of $r + 1$ elements $\{h_i\}$ of $\mathcal{U}_{K/k}$ such that

$$\mathcal{U}_{K/k} / \langle h_i \rangle_{\mathbb{Z}[\text{Gal}(K/k)]}$$

is finite with minimal index.

We are going to show that a fundamental system of units exists. Instead of working directly with $\mathcal{U}_{K/k}$, we find it more convenient to formalize a more general situation and then restrict to the specific setting above. We begin with any additive commutative group G playing the role of $\mathcal{U}_{K/k}$, which we also assume is a \mathcal{O}_K -module (for $\mathcal{U}_{K/k}$ the action will be given by $\zeta \mapsto \sigma$). Note that the natural notation for $\mathcal{U}_{K/k}$ is multiplicative rather than additive, but to use the language of modules we are forced to work with additive groups. Since we need to use the element $\zeta \in \mathcal{O}_K$ several times, we decided to work in this section with a different model of the ring of integers of K [is](#):

```

def CyclotomicIntegers : Type := AdjoinRoot (cyclotomic p ℤ)
local notation "A" => (CyclotomicIntegers p) -- This is just ℤ[ζ]

```

This means that A is $\mathbb{Z}[\alpha]$, where α is an unspecified root of the p -th cyclotomic polynomial (of course this ring is abstractly isomorphic to \mathcal{O}_K). We start by defining the notion of *system of s units* where s is a natural number. It is just a set of s elements of G that are linearly independent over A .

```
structure systemOfUnits (G : Type*) [AddCommGroup G] (s : ℕ) where
  units : Fin s → G --A choice of s elements of G
  linearIndependent : LinearIndependent A units
```

We prove that if G is free as \mathbb{Z} -module of rank $s(p-1)$, than a system of s units always exists.

We now move on to the definition of *fundamental systems of r units*. We start by introducing the notion of *maximal system of s units*.

```
abbrev systemOfUnits.IsMaximal {s : ℕ} {p : ℕ+} {G : Type*} [AddCommGroup G]
  [Module A G] (sys : systemOfUnits (G := G) p s) :=
  Fintype (G / Submodule.span A (Set.range sys.units))
```

By definition, being maximal means that the quotient by the submodule generated by the elements of the system of units is finite. We then prove that if there is a system of s units then there is a maximal one if $\text{rk}_{\mathbb{Z}}(G) = s(p-1)$.

A system of s units is *fundamental* if it is maximal and the submodule generated by the elements of the system has index smaller than those generated by any other maximal system.

```
def systemOfUnits.IsFundamental [Module A G] (h : systemOfUnits p G s) :=
  ∃ _ : h.IsMaximal, ∀ (S : systemOfUnits p G s) (_ : S.IsMaximal), h.index ≤ S.index
```

We then prove that a fundamental system of s units always exists if G is free as \mathbb{Z} -module of rank $s(p-1)$. The key property of fundamental systems of units is the following

Lemma 7.3. *Let a_0, \dots, a_{s-1} be integers not all divisible by p and $\{h_i\}$ a fundamental system of s units, then*

$$\prod_{i=0}^{s-1} h_i^{a_i} \neq u/\sigma(u)$$

for any $u \in \mathcal{U}_{K/k}$.

Our formalization is (note the additive notation)

```
lemma corollary [Module A G] (S : systemOfUnits p G s) (hs : S.IsFundamental)
  (a : Fin s → ℤ) (ha : ∃ i, ¬ (p : ℤ) ∣ a i) :
  ∀ g : G, (1 - zeta p) · g ≠ ∑ i, a i · S.units i := by ...
```

Here $\text{zeta} : \text{CyclotomicIntegers } p := \text{AdjoinRoot.root } _$ is the fixed root of the p -th cyclotomic polynomial in A , that corresponds to the primitive p -th root of unity ζ (and that acts as the fixed generator σ).

7.2. Hilbert 91. We now go back to our specific situation of a cyclic field extension K/k . We want to show that a system of fundamental $r+1$ units exists in the case $r = \text{rk}_{\mathbb{Z}}(\mathcal{O}_k^\times)$ and $G = \mathcal{U}_{K/k}$. As $\mathcal{U}_{K/k}$ is torsion-free and hence free, this amounts to show that its rank is $(r+1)(p-1)$. This is essentially the content of Hilbert's Theorem 91. We begin by defining the group $\mathcal{U}'_{K/k} = \mathcal{O}_K^\times / \mathcal{O}_k^\times$ as

```
def RelativeUnits (k K : Type*) [Field k] [Field K] [Algebra k K] :=
  ((ℕ K) × / (MonoidHom.range <|
    Units.map (algebraMap (ℕ k) (ℕ K) : (ℕ k) →* (ℕ K))))
```

Here $\text{MonoidHom.range } <| \text{Units.map (algebraMap (ℕ k) (ℕ K) : (ℕ k) →* (ℕ K))}$ denotes the image of the units in k under the natural embedding into \mathcal{O}_K^\times . Since we are going to work with a fixed generator of the Galois group, we find it more convenient to package together all the data we have

```
def relativeUnitsWithGenerator (_hp : Nat.Prime p) (_hKL : finrank k K = p)
  (σ : K ≃_a [k] K) (_hσ : ∀ x, x ∈ Subgroup.zpowers σ) : Type _ :=
  RelativeUnits k K
```

This is the same as `RelativeUnits`, but it contains the choice of a generator σ . Finally, we define our additive torsion-free group $\mathcal{U}_{K/k} = \mathcal{U}'_{K/k} / \mathcal{U}'_{K/k, \text{tors}}$ as

```

local notation "G" =>
  Additive (relativeUnitsWithGenerator p hp hKL σ hσ) /
  AddCommGroup.torsion (Additive (relativeUnitsWithGenerator p hp hKL σ hσ))

```

Here, if \mathfrak{H} is a multiplicative group, `Additive \mathfrak{H}` is the same group, but with additive notation. As explained above we are forced to use it, and it causes a little bit of friction, but all the relevant results to go from \mathfrak{H} to `Additive \mathfrak{H}` were already in MATHLIB. We show [↗](#) that if we consider G as a $\mathbb{Z}[X]$ -module via $X \mapsto \sigma$, then it is torsion with respect to the subgroup generated by the cyclotomic polynomial. This endows [↗](#) G with a structure of an A -module. Moreover, we show [↗](#) that it has rank (as a \mathbb{Z} -module) is equal to $(r+1)(p-1)$ (this step is rather delicate: we need first of all to take care of the quotient by the torsion submodule and then we use that our extension is unramified at infinite places since p is assumed to be odd).

Putting it all together, we can use our general existence result above to show that there is in fact a fundamental system of $r+1$ units for G as claimed by Hilbert's Theorem 91 [↗](#):

```

lemma Hilbert91 :
  ∃ S : systemOfUnits p G (NumberField.Units.rank k + 1), S.IsFundamental :=
  systemOfUnits.IsFundamental.existence p hp G
  (NumberField.Units.rank k + 1) (finrank_G p hp hKL σ hσ)

```

7.3. Proving Hilbert's Theorem 92. We can now finish the proof of Hilbert's Theorem 92, reducing the proof of Fermat's Last Theorem for regular primes to Hilbert's Theorem 90.

Let K/k be as in 7.1. Remember that in particular p is an odd prime and that σ is a generator of $\text{Gal}(K/k)$. Recall that $r = \text{rk}_{\mathbb{Z}}(\mathcal{O}_k^\times)$ and let h_i be a fundamental system of $r+1$ units for $\mathcal{U}_{K/k}$, that exists by Hilbert's Theorem 91 above. We denote by H_i fixed lifts to elements of \mathcal{O}_K^\times . We fix $h \in \mathbb{N}$ be such that k contains a p^h -th root of unity ν but no p^{h+1} -th root (note we allow h to be zero). Let $\xi = \nu^{p^{h-1}}$ which is now a p -th root of unity as above, with the convention that if $h = 0$ we have $\xi = \nu = 1$.⁴

With these notations the proof proceeds as follows (in the formalization most of the statements are actually more general than those below, for example stated for any family of r elements of \mathcal{O}_K^\times when the fact that being a fundamental system of r units is not needed):

- (1) By Hilbert's Theorem 90, Theorem 8.1 below, we know that, since ξ has norm 1, we can write $\xi = \varepsilon/\sigma(\varepsilon)$ for some $\varepsilon \in \mathcal{O}_K$.
- (2) We can assume that there is some $E \in \mathcal{O}_K^\times$ such that

$$\xi = E/\sigma(E)$$

as otherwise we could take ξ as the element required by 7.1.

- (3) Note that from $\xi = E/\sigma(E)$ we have that [↗](#) $\text{Norm}_{K/k}(E) = E^p$.
- (4) Let $\eta_i = \text{Norm}_{K/k}(H_i)$ for $i \in 0, \dots, r$ and let $\eta_{r+1} := \text{Norm}_{K/k}(E) = E^p$ (again most of the formal statements are for general η_i).
- (5) There exist [↗](#) $a, a_i \in \mathbb{Z}$ for $i \in 0, \dots, r+1$ such that

$$\prod_{i=0}^{r+1} \eta_i^{a_i} = \nu^{ap}$$

with a_{i_0} not divisible by p for some i_0 . Moreover, if $\nu = 1$ (i.e. $h = 0$) then we can take i_0 to be different from $r+1$.

- (6) With these exponents a_i in hand we can now construct our element. Set [↗](#)

$$J := \nu^{-a} \prod_{i=0}^{r+1} H_i^{a_i}.$$

where we set $H_{r+1} = E$. We need to show it satisfies the required properties.

- (7) Taking its norm we have that [↗](#)

$$\text{Norm}_{K/k}(J) = \left(\prod_{i=0}^{r+1} \eta_i^{a_i} \right) \nu^{-ap}$$

which is 1 by our the condition on the a_i 's in (5).

⁴In the formalization this convention is automatic, as $0-1=0$ for natural numbers in LEAN.

- (8) Now if J was of the form $\varepsilon/\sigma(\varepsilon)$ for some unit ε , then the same would be true of its image in $\mathcal{U}_{K/k}$. Its image is

$$\nu^{-a} \prod_{i=0}^r h_i^{a_i} \cdot E^{a_{r+1}}.$$

Now, since $\nu \in \mathcal{O}_k^\times$, its image in the quotient is 1 (remember that here we are using multiplicative notation). Moreover, since $E^p = \text{Norm}_{K/k}(E)$ also belongs to \mathcal{O}_k^\times we have that its image is 1, so the image of E is torsion. Being $\mathcal{U}_{K/k}$ torsion-free, we have that E becomes equal to 1 in the quotient and the image of J is

$$\prod_{i=0}^r h_i^{a_i}.$$

- (9) By Lemma 7.3 to conclude the proof it is enough to show that not all of the a_i for $i \in \{0, \dots, a_r\}$ are divisible by p . Assume for contradiction this is the case, then a_{r+1} must be divisible by p . Now, this means $h \neq 0$ since otherwise this would contradict our choice of a_i 's. But now this means that $\eta_{r+1} = \text{Norm}_{K/k}(E) = E^p$ is the p -th power of a unit in k (this follows from (5)). In particular E is a *unit* now in k . But this means that $\sigma(E) = E$, in particular $\xi = 1$, but this cannot be the case as we have already shown $h \neq 0$.

The proof of the last two points are inlined in the proof of [almostHilbert92](#), that is Hilbert's Theorem 92 with the additional assumption that K/k is unramified at all infinite places.

Remark 7.4. Note that we have slightly simplified the proof of Hilbert's Theorem 92 as found in [HLA+98], which splits into two cases, the first when K does not contain an p -th root and the second where it contains a p^h -th root (but not a p^{h+1} -th root) for some h . Our proof instead allows $h = 0$ and so follows the second case.

8. HILBERT'S THEOREM 90

Hilbert's Theorem 90 is a nowadays classical result about Galois cohomology. MATHLIB already contains a version of (a generalization of) it, thanks to the work of Amelia Livingston, see [Liv23]. We explain in this section the precise statement we need, and how to get there given what already is in MATHLIB.

The statement we needed above is the following, matching Hilbert's original formulation.

Theorem 8.1 (Hilbert's Theorem 90, concrete version). *Let L/K be a cyclic extension of fields and let $\sigma \in \text{Gal}(L/K)$ be a generator. If $\eta \in L$ is such that $\text{Norm}_{L/K}(\eta) = 1$, then there exists an $\varepsilon \in \mathcal{O}_L$ such that $\varepsilon \neq 0$ and $\eta\sigma(\varepsilon) = \varepsilon$.*

Proof. By clearing denominators, it is enough to prove that there is $\varepsilon \in L$ such that $\varepsilon \neq 0$ and $\eta = \varepsilon/\sigma(\varepsilon)$. We then deduce the theorem from Noether's generalization of Hilbert's Theorem 90, Theorem 8.2 below. Let $c: \text{Gal}(L/K) \rightarrow L^\times$ be the function $\tau \mapsto \prod_{i=0}^n \sigma^i(\eta)$, where $n \in \mathbb{N}$ is the unique natural number smaller than the order of $\text{Gal}(L/K)$ such that $\sigma^n = \tau$. Using that $\text{Norm}_{L/K}(\eta) = \prod_{\tau \in \text{Gal}(L/K)} \tau(\eta) = 1$ we have that c is a cocycle, i.e. $c(\tau_1\tau_2) = \tau_1(c(\tau_2))c(\tau_1)$. Since $H^1(G, L^\times)$ is trivial by Theorem 8.2, we have that c is a coboundary. By definition this means that there is $x \in L$ such that $c(\tau) = \tau(x)/x$ for all $\tau \in \text{Gal}(L/K)$ and in particular $c(\sigma) = \sigma(x)/x$. Since $c(\sigma) = \eta$, we have that $x\eta = \sigma(x)$ and $\varepsilon = x^{-1}$ satisfies the conditions of the theorem. \square

Here is our formalization [↗](#)

```
variable {K L : Type*} [Field K] [Field L] [Algebra K L]
[IsGalois K L] [FiniteDimensional K L]
{A B : Type*} [CommRing A] [CommRing B] [Algebra A B] [Algebra A L] [Algebra A K]
[Algebra B L] [IsScalarTower A B L] [IsScalarTower A K L] [IsFractionRing A K]
[IsDomain A] [IsIntegralClosure B A L] [IsDomain B]

lemma Hilbert90_integral (σ : L ≃_α [K] L) (hσ : ∀ x, x ∈ Subgroup.zpowers σ)
(η : B) (hη : Algebra.norm K (algebraMap B L η) = 1) :
∃ ε : B, ε ≠ 0 ∧ η * galRestrict A K L B σ ε = ε := by ...
```

Note that in the formalization, instead of working with \mathcal{O}_K and \mathcal{O}_L (the integral closure of \mathbb{Z} in K and L respectively), we decided to work with general integral domains A and B such that B is in the integral closure of A in L ; see Section 9 for more details about this design decision.

The formal proof follows the informal one, clearing denominators (a step that in LEAN is not completely trivial since we need to consider the restriction of σ to B) and deducing the theorem from [\[Liv23\]](#)

```
lemma Hilbert90 {η : L} (hη : Algebra.norm K η = 1) :
  ∃ ε : L, η = ε / σ ε := by ...
```

Note that even if technically the statement does not contain the fact that $\varepsilon \neq 0$, this is automatic: if $\varepsilon = 0$, then $\sigma(\varepsilon) = 0$ and hence $\varepsilon/\sigma(\varepsilon) = 0$ (remember that in LEAN division by 0 returns 0), so $\eta = 0$, that is impossible because $\text{Norm}_{L/K}(\eta) = 1$.

Everything is now reduced to the following

Theorem 8.2 (Noether’s generalization of Hilbert’s Theorem 90). *Let L/K be a Galois extension of fields with Galois group G . Then $H^1(G, L^\times)$ is trivial.*

This is a nowadays classical result in group cohomology, and we will not recall the proof. Thanks to the formalization project [\[Liv23\]](#) by Amelia Livingston, it is already present in MATHLIB [\[Liv23\]](#)

```
-- Noether's generalization of Hilbert's Theorem 90: given a finite extension of fields and a
function 'f : Aut_K(L) → L^×' satisfying 'f(gh) = g(f(h)) * f(g)' for all 'g, h : Aut_K(L)',
there exists 'β : L^×' such that 'g(β)/β = f(g)' for all 'g : Aut_K(L)'. -/
theorem isMulOneCoboundary_of_isMulOneCocycle_of_aut_to_units
(f : (L ≃_a [K] L) → L^×) (hf : IsMulOneCocycle f) :
  IsMulOneCoboundary f := by ...
```

here `IsMulOneCocycle` and `IsMulOneCoboundary` are defined as follows [\[Liv23\]](#) (where M is any G module, for a group G)

```
def IsMulOneCocycle (f : G → M) : Prop := ∀ g h : G, f (g * h) = g · f h * f g

def IsMulOneCoboundary (f : G → M) : Prop := ∃ x : M, ∀ g : G, g · x / x = f g
```

Note that group cohomology here is defined in very concrete terms, as cocycles modulo coboundaries, and not via an abstract machinery like derived functors (that exist in MATHLIB). Even if at some point MATHLIB will need the connection between this two definitions, the current design choice is very convenient for us, since it allowed to prove the explicit theorem we need quickly.

9. IMPLEMENTATION ISSUES

We now describe two implementation issues we encountered during our formalization project. As the problem of the coercions from \mathcal{O}_K^\times to \mathcal{O}_K to K and the diamonds related characteristic zero fields are already discussed in [\[BBBRB23, Section 3\]](#), we will not discuss those.

After the port to LEAN4, we noticed that the project was very slow in several places, making almost impossible to progress with the formalization. In collaboration with the MATHLIB community (especially Matthew Ballard, Kevin Buzzard and Floris van Doorn), we identified that the main bottleneck was the definition of the ring of integers of a number field in MATHLIB, that was

```
def RingOfIntegers := integralClosure ℤ K
@[inherit_doc] scoped notation "O" => NumberField.ringOfIntegers
```

In general, if A is an R -algebra, the type of `integralClosure R A` is `Subalgebra R A`. This has the advantage that a lot of instances, for example `CommRing (O K)`, are found automatically by LEAN. On the other hand, it causes the following drawback. MATHLIB contains the following instance [\[Liv23\]](#), where α is a type endowed with `SMul A α`, that is a scalar multiplication by A .

```
instance [SMul A α] (S : Subalgebra R A) : SMul S α := ...
```

Mathematically this simply means that if we know how to multiply elements of α by any $(a : A)$ we can also multiply elements of α by any $(s : S)$. In particular, any time an instance like `SMul (O K) (O K)` is needed (for example looking for `Module (O K) (O K)`), LEAN will look for an instance of `SMul K (O K)` (a mathematically meaningless problem), and this search will of course fail. What we realised is that this search is rather slow to fail, and, due to the intricacies of the algebra hierarchy in MATHLIB, it is performed many times. We tried to manually lower the priority of the above instance, but this did not solve the problem. Indeed it is sometimes difficult to control the path chosen by the typeclass inference

system, and this solution would not scale anyway. In the pull request [#12386](#) we then decided to change the definition of `ringOfIntegers` to [↗](#)

```
def RingOfIntegers : Type _ := integralClosure ℤ K
```

The only difference with the above one is that the type of \mathcal{O}_K is now `Type _` rather than `Subalgebra ℤ K`. To be precise, \mathcal{O}_K is now the underlying type of `integralClosure ℤ K`: this means that, for example, the following instance [↗](#)

```
instance : CommRing (℧ K) := inferInstanceAs (CommRing (integralClosure _ _))
```

has to be added by hand (here the command `inferInstanceAs` allows LEAN to see through the definition of `RingOfIntegers`). On the other hand, it also means that the instance from `SMul K (℧ K)` to `SMul (℧ K) (℧ K)` is not even considered when solving `Module (℧ K) (℧ K)`. This simple modification made essentially all MATHLIB’s file related to ring of integers of number fields much faster (see the [benchmark results](#)) and drastically improved the speed of FLT-REGULAR. Note that the number of instances that have to be added by hand is quite limited, and the gain in speed is huge: we expect that a similar modification will be needed in MATHLIB in the future for other analogous situations, for example in the definition of the adèle ring of a global field.

Another issue is related to field extensions: if L/K is such an extension, then \mathcal{O}_L is naturally an \mathcal{O}_K -algebra. If moreover K and L are number fields, it is easy to prove that mathematically \mathcal{O}_L is the integral closure of \mathcal{O}_K in L . In LEAN this equality (of types!) is not provable, and the formal translation of this results is the datum of an isomorphism (as \mathcal{O}_K -algebras) between \mathcal{O}_L and the integral closure of \mathcal{O}_K in L . Working with such an isomorphism can be annoying, as it is going to appear several times. To avoid this issue, we decided to work more generally in the so called *AKLB setting*:

```
variable {A K L B : Type*} [CommRing A] [CommRing B] [Field K] [Field L]
  [Algebra A B] [Algebra A L] [Algebra A K] [Algebra B L]
  [IsScalarTower A B L] [IsScalarTower A K L] [IsFractionRing A K] [IsIntegralClosure B A L]
```

Any result holding in this setting will apply both to $B = \mathcal{O}_L$ and to $B = \text{integralClosure } (\mathcal{O}_K) L$, allowing very often to avoid the use of the isomorphism above (one may assume `IsIntegralClosure ℤ A K` to state that A “is” \mathcal{O}_K). This setting is very common in algebraic number theory, and we believe this way of formalizing it should be used in MATHLIB every time it is possible.

10. FUTURE WORK

As explained in [\[BBBRB23, Section 4\]](#), one peculiar aspect of our project is that the integration to MATHLIB is happening in real time. Even if we don’t think the full proof of Fermat’s Last Theorem for regular primes should be included into MATHLIB (as is it very technical and mathematically superseded by more modern approaches), most of the prerequisites we formalized are fundamental results in algebraic number theory and should move to the main library. The process of opening pull requests from FLT-REGULAR to MATHLIB never stopped, and we expect that all the relevant material will end up in MATHLIB at some point. One notable example of this process is the proof of Fermat’s Last Theorem for $n = 3$, that is now in MATHLIB [↗](#). Note that this case has to be done by hand, even when taking into account modern proofs and it is rather non trivial (for example it is more difficult than the case $n = 4$, also in MATHLIB [↗](#)). In particular, this case will be need for Kevin Buzzard’s [project](#) of formalizing a proof of the full Fermat’s Last Theorem.


Concerning future formalizations, one natural question that is left by the current status of FLT-REGULAR is that the condition for a prime of being regular is rather abstract, and difficult to be proved in practice. Indeed, the computation of the class number of $\mathbb{Q}(e^{\frac{2\pi i}{n}})$ is a difficult problem, even on paper. Thanks to Xavier Roblot’s work, Minkowski’s bound is in MATHLIB [↗](#). In particular it is easy to prove [↗↗](#) that both $\mathbb{Z}[e^{\frac{2\pi i}{3}}]$ and $\mathbb{Z}[e^{\frac{2\pi i}{5}}]$ are principal ideal domains (since the Minkowski’s bound is 1 in this cases), and hence 3 and 5 are regular primes [↗↗](#) (we also know that 2 is regular [↗](#)) and the case $n = 5$ of Fermat’s Last Theorem follows [↗](#). This is to our knowledge the first formalization of the nonexistence of nontrivial solutions to

$$x^5 + y^5 = z^5$$

A natural question is to give more explicit examples of regular primes. One can prove that $\mathbb{Z}[e^{\frac{2\pi i}{p}}]$ is principal for all $p \leq 19$ (the converse also holds), but the proof is more and more involved: indeed Minkowski’s bound becomes exponentially larger and one has to check a lot of cases by hand (on the

other hand we think that the case $p = 7$ should be doable without too much trouble). Note that it is known that $\mathbb{Z}[e^{\frac{2\pi i}{p}}]$ a principal ideal domain if and only if it is a Euclidean domain (and for $p \leq 13$ this is equivalent to be norm-Euclidean), so in principle one can prove regularity (at least for $p \leq 13$) via an explicit computation, but again this is very cumbersome and in any case it cannot work for $p > 19$ (for example $\mathbb{Z}[e^{\frac{2\pi i}{23}}]$ has class number 3, so 23 is regular). The modern approach to prove regularity is via the following (see [Was97, Theorem 5.34])

Theorem 10.1. *An odd prime p is regular if and only if p does not divide the denominator of the Bernoulli number B_n for $n = 2, 4, \dots, p - 3$.*

Since Bernoulli numbers are very easy to compute (and already in MATHLIB ) this criterion gives a very easy way of checking whether a given p is regular (for example one immediately obtains that the only irregular primes $p \leq 100$ are 37, 59 and 67). The proof of Theorem 10.1 requires the p -adic class number formula, that is proved using p -adic L -functions. The basics theory of p -adic L -functions have been formalized by Narayanan in LEAN3 (see [Nar23]), and a port to LEAN4 is work in progress. For these reasons we believe that the formalization of Theorem 10.1 is within reach, and we plan to accomplish it in the near future.

REFERENCES

- [BBBRB23] Alex J. Best, Christopher Birkbeck, Riccardo Brasca, and Eric Rodriguez Boidi, *Fermat’s Last Theorem for Regular Primes*, 14th International Conference on Interactive Theorem Proving (ITP 2023) (Dagstuhl, Germany) (Adam Naumowicz and René Thiemann, eds.), Leibniz International Proceedings in Informatics (LIPIcs), vol. 268, Schloss Dagstuhl – Leibniz-Zentrum für Informatik, 2023, pp. 36:1–36:8.
- [BDNdC22] Anne Baanen, Sander R. Dahmen, Ashvni Narayanan, and Filippo A. E. Nuccio Mortarino Majno di Capriglio, *A Formalization of Dedekind Domains and Class Groups of Global Fields*, J. Autom. Reason. **66** (2022), no. 4, 611–637.
- [BS66] Z. I. Borevich and I. R. Shafarevich, *Number theory. Translated by Newcomb Greenleaf*, Pure Appl. Math., Academic Press, vol. 20, New York and London: Academic Press, 1966 (English).
- [HLA⁺98] D. Hilbert, F. Lemmermeyer, I.T. Adamson, N. Schappacher, and R. Schoof, *The theory of algebraic number fields*, Springer Berlin Heidelberg, 1998.
- [Liv23] Amelia Livingston, *Group cohomology in the lean community library*, International Conference on Interactive Theorem Proving, 2023.
- [mC20] The mathlib Community, *The Lean Mathematical Library*, Proceedings of the 9th ACM SIGPLAN International Conference on Certified Programs and Proofs (New York, NY, USA), CPP 2020, Association for Computing Machinery, 2020, p. 367–381.
- [Nar23] Ashvni Narayanan, *Formalization of p -adic l -functions in lean 3*, 2023.
- [SD01] H. P. F. Swinnerton-Dyer, *A brief guide to algebraic number theory*, London Mathematical Society Student Texts, vol. 50, Cambridge University Press, Cambridge, 2001. MR 1826558
- [TW95] Richard Taylor and Andrew Wiles, *Ring-theoretic properties of certain Hecke algebras*, Ann. of Math. (2) **141** (1995), no. 3, 553–572. MR 1333036
- [Was97] Lawrence C. Washington, *Introduction to cyclotomic fields*, second ed., Graduate Texts in Mathematics, vol. 83, Springer-Verlag, New York, 1997. MR 1421575
- [Wie23] Eric Wieser, *Multiple-inheritance hazards in dependently-typed algebraic hierarchies*, Intelligent Computer Mathematics (Cham) (Catherine Dubois and Manfred Kerber, eds.), Springer Nature Switzerland, 2023, pp. 222–236.
- [Wil95] Andrew Wiles, *Modular elliptic curves and Fermat’s last theorem*, Ann. of Math. (2) **141** (1995), no. 3, 443–551. MR 1333035

URL: <https://alexjbest.github.io/>
Email address: alexjbest@gmail.com

UNIVERSITY OF EAST ANGLIA
Email address: c.birkbeck@uea.ac.uk
URL: <https://cdbirkbeck.wixsite.com/website>

INSTITUT DE MATHÉMATIQUES DE JUSSIEU-PARIS RIVE GAUCHE, UNIVERSITÉ PARIS-CITÉ
Email address: riccardo.brasca@imj-prg.fr
URL: <https://webusers.imj-prg.fr/~riccardo.brasca/>

Email address: eric.rodriguez-boidi@kcl.ac.uk

Email address: ruben.vandavelde@gmail.com

Email address: a.yang24@imperial.ac.uk